

Nemanjici Ontologija

Projektna dokumentacija

Predmet:	Arhitektura softvera za upravljanje pametnim mrežama (MAS)
Tema:	RDF/SPARQL ontologija srpske srednjovekovne dinastije Nemanjica
Alati:	Enterprise Architect, SPARQL Playground, DBpedia, Wikidata
Format:	RDF/XML + Turtle (.ttl) + SPARQL (.rq)
Radio:	Milorad Maksic

1. Pregled projekta

Ovaj projekat predstavlja RDF/SPARQL ontologiju srpske srednjovekovne dinastije Nemanjica. Dinastija Nemanjica vladala je Srbijom od 1166. do 1371. godine i predstavlja jedan od najznacajnijih perioda srpske istorije. Ontologija modeluje vladare, crkve, teritorije, drzave, bitke, pravne dokumente i druge entitete iz tog perioda.

Struktura projekta

Projekat je organizovan prema standardnoj strukturi koja prati primjer Garden projekta:

Folder/Fajl	Opis
Model/	EA dijagram, TTL schema i RDF/XML fajl
Model/Nemanjici.qea	Enterprise Architect projekat (UML dijagram)
Model/Nemanjici.ttl	Ontologija u Turtle formatu (kopija iz schema)
Model/NemanjiciSchema-rdfs-augmented.xml	RDF/XML schema eksportovana iz EA
Model/Nemanjici.png	Screenshot UML dijagrama iz EA
Ttl/	TTL fajlovi i SPARQL upiti
Ttl/ttl-data/ontology.ttl	Definicija klasa i svojstava (schema)
Ttl/ttl-data/resource.ttl	Instance podataka (vladari, crkve, bitke...)
Ttl/ttl-data/prefixes.ttl	Prefiksi koji se koriste u projektu
Ttl/queries/01.rq - 10.rq	10 lokalnih SPARQL upita
Ttl/queries/11.rq - 20.rq	10 vanjskih SPARQL upita (Wikidata)

2. UML dijagram u Enterprise Architect

UML class dijagram je kreiran u Enterprise Architect (EA) alatu. Dijagram prikazuje sve klase, njihove attribute, veze izmedju klasa i enumeracije. EA je koriscen za vizuelno modelovanje ontologije prije nego sto su generisani TTL fajlovi.

Proces kreiranja dijagrama

1. Kreiran novi EA projekat (File -> New Project) i sacuvan kao Nemanjici.qea
2. Kreiranje potrebnih klasa za ontologiju
3. Klase prevucene na Class Diagram (drag & drop iz Browser panela)
4. Asocijacije nacrtane rukno izmedju klasa koriscenjem hover strelica
5. Dijagram uredjen i sacuvan kao PNG (Publish -> Save -> Save as Image)

Klase u modelu

Klasa	Opis	Nasljedjuje
Osoba	Bazna klasa za sve osobe	-
Vladar	Vladari Srbije sa titulama i periodom vladavine	Osoba
Svestenik	Crkveni funkcioneri	Osoba
Plemic	Plemici i oblasni gospodari	Osoba
Dinastija	Vladajuca porodica (Nemanjici)	-
Drzava	Drzavna tvorevina (zupanija, kraljevina, carstvo)	-
Teritorija	Geografska oblast kojom je vladano	-
Bitka	Vojni sukob sa datumom i lokacijom	-
PravniDokument	Zakoni i povelje	-
Ugovor	Medjudrzavni ugovori	-
Crkva	Manastiri i crkve	-
Grad	Gradovi sa info o prestonicastvu	-

Enumeracije

Enumeracija	Vrijednosti
TipVladara	VelikaZupan, Kralj, Car, Knez, Despot
TipTeritorije	Zupanija, Kraljevina, Carstvo, Oblast, Zupa
TipCrkve	Manastir, Crkva

M:N relacija

Kljucna M:N relacija u modelu je izmedju klasa Vladar i Teritorija. Jedan vladar je mogao vladati vise teritorija, a jednom teritorijom su mogli vladati vise vladara u razlicitim periodima. Ova relacija je implementirana kroz svojstvo Vladar.vladaoTeritorijom (sa inverznim Teritorija.vladao).

3. TTL fajlovi (Turtle format)

3.1 prefixes.ttl

Definise prefikse koji se koriste u svim ostalim fajlovima:

```
@prefix nemanjicischema: <http://iec.ch/TC57/NemanjiciSchema#> .
@prefix xsd: <http://www.w3.org/2001/XMLSchema#> .
```

3.2 ontology.ttl

Definise sve klase i njihova svojstva. Koristi ext: prefiks za multiplicity i stereotype, sto je kompatibilno sa Garden projektom. Primjer definicije klase Vladar:

```
nemanjicischema:Vladar a rdfs:Class ;
    rdfs:label "#Vladar" ;
    rdfs:subClassOf nemanjicischema:Osoba .

nemanjicischema:Vladar.titula a rdf:Property ;
    rdfs:label "titula" ;
    rdfs:domain nemanjicischema:Vladar ;
    rdfs:range nemanjicischema:TipVladara ;
    ext:multiplicity ext:M:1 ;
    ext:stereotype uml:attribute .
```

3.3 resource.ttl

Sadrzi sve instance (konkretne podatke). Vrijednosti enumeracija su stringovi, sto je uskladjeno sa Garden projektom. Primjer instance vladara:

```
nemanjicischema:stefan_dusan a nemanjicischema:Vladar ;
    nemanjicischema:Osoba.ime "Stefan Dusan" ;
    nemanjicischema:Osoba.godinaRodjenja 1308 ;
    nemanjicischema:Osoba.godinaSmrti 1355 ;
    nemanjicischema:Vladar.titula "Car" ;
    nemanjicischema:Vladar.vladaoOd 1331 ;
    nemanjicischema:Vladar.vladaoDo 1355 ;
    nemanjicischema:Vladar.vladaoDrzavom nemanjicischema:srpsko_carstvo ;
    nemanjicischema:Vladar.vladaoTeritorijom nemanjicischema:ter_raska ,
        nemanjicischema:ter_kosovo , nemanjicischema:ter_macedonia .
```

Instance u resource.ttl

Tip	Instance	Broj
Vladar	Stefan Nemanja, Stefan Prvovencani, Stefan Radoslav, Stefan Vladislav, Stefan Uros I, Stefan Dragutin, Stefan Milutin, Stefan Decanski, Stefan Dusan, Stefan Uros V	10
Svestenik	Sveti Sava, Patrijarh Joanikije II	2
Plemic	Vuk Brankovic, Lazar Hrebeljanovic	2
Crkva	Studenica, Hilandar, Zica, Gracanica, Decani, Pecka Patrijarsija	6
Teritorija	Raska, Zeta, Kosovo, Makedonija, Hum	5
Drzava	Srpska Velika Zupanija, Kraljevina Srbija, Srpsko Carstvo	3
Bitka	Bitka na Velbudu, Osvajanje Skoplja, Bitka na Marici	3
PravniDokument	Dusanov Zakonik, Povelja Stefana Nemanje	2

Ugovor	Ugovor sa Vizantijom, Ugovor sa Mleccanima	2
Grad	Ras, Skoplje, Prizren	3
Dinastija	Nemanjici	1

4. Lokalni SPARQL upiti (01.rq - 10.rq)

Lokalnih 10 upita radi nad TTL fajlovima (ontology.ttl + resource.ttl) u SPARQL Playgroundu. Pokrecu se na lokalno ucitanim podacima - ne zahtijevaju internet konekciju.

Upit 01.rq: Svi vladari sa titulama

Opis: Vraca sve vladare sortirane po godini pocetka vladavine. Prikazuje ime, titulu i period vladavine.

Ocekivani rezultat: 10 redova (svi vladari od Stefana Nemanje do Stefana Urosa V)

```
SELECT ?ime ?titula ?od ?do WHERE {  
    ?v a nemanjicischema:Vladar ;  
    nemanjicischema:Osoba.ime ?ime ;  
    nemanjicischema:Vladar.titula ?titula ;  
    nemanjicischema:Vladar.vladaoOd ?od ;  
    nemanjicischema:Vladar.vladaoDo ?do .  
}  
ORDER BY ?od
```

ime	titula	od	do
"Stefan Nemanja"	"VelikaŽupan"	"1166"	"1196"
"Stefan Prvovencani"	"Kralj"	"1196"	"1228"
"Stefan Radoslav"	"Kralj"	"1228"	"1233"
"Stefan Vladislav"	"Kralj"	"1233"	"1243"
"Stefan Uros I"	"Kralj"	"1243"	"1276"
"Stefan Dragutin"	"Kralj"	"1276"	"1282"
"Stefan Milutin"	"Kralj"	"1282"	"1321"
"Stefan Decanski"	"Kralj"	"1322"	"1331"
"Stefan Dusan"	"Car"	"1331"	"1355"
"Stefan Uros V"	"Car"	"1355"	"1371"

Upit 02.rq: Manastiri sa ktitorima

Opis: Spaja klase Crkva i Vladar da dobije ime ktitora (graditelja) svakog manastira. Filtrira samo tip 'Manastir'.

Ocekivani rezultat: 5 manastira (Studenica, Hilandar, Zica, Gracanica, Decani)

```
SELECT ?manastir ?ktitorIme ?godina WHERE {  
    ?c a nemanjicischema:Crkva ;  
    nemanjicischema:Crkva.naziv ?manastir ;  
    nemanjicischema:Crkva.tipCrkve ?tip ;  
    nemanjicischema:Crkva.ktitor ?ktitor .  
    ?ktitor nemanjicischema:Osoba.ime ?ktitorIme .  
    FILTER (?tip = "Manastir")  
}  
ORDER BY ?godina
```

manastir	ktitorIme	godina
"Manastir Studenica"	"Stefan Nemanja"	"1190"
"Manastir Hilandar"	"Stefan Nemanja"	"1198"
"Manastir Zica"	"Stefan Prvovencani"	"1208"
"Manastir Gracanica"	"Stefan Milutin"	"1321"
"Manastir Decani"	"Stefan Decanski"	"1327"

Upit 03.rq: Vladari cije ime sadrzi Stefan

Opis: Koristi FILTER sa CONTAINS funkcijom za pretragu vladara po imenu. Demonstrira string filtriranje.

Ocekivani rezultat: 9 vladara (svi sem Lazara Hrebeljanovica koji je plemic)

```
SELECT ?ime ?titula WHERE {
  ?v a nemanjicischema:Vladar ;
    nemanjicischema:Osoba.ime ?ime ;
    nemanjicischema:Vladar.titula ?titula .
  FILTER (CONTAINS(?ime, "Stefan"))
} ORDER BY ?od
```

ime	titula	od	do
"Stefan Nemanja"	"VelikaZupan"	"1166"	"1196"
"Stefan Prvovencani"	"Kralj"	"1196"	"1228"
"Stefan Radoslav"	"Kralj"	"1228"	"1233"
"Stefan Vladislav"	"Kralj"	"1233"	"1243"
"Stefan Uros I"	"Kralj"	"1243"	"1276"
"Stefan Dragutin"	"Kralj"	"1276"	"1282"
"Stefan Milutin"	"Kralj"	"1282"	"1321"
"Stefan Decanski"	"Kralj"	"1322"	"1331"
"Stefan Dusan"	"Car"	"1331"	"1355"
"Stefan Uros V"	"Car"	"1355"	"1371"

Upit 04.rq: Broj crkava po vladaru

Opis: Koristi GROUP BY i COUNT agregaciju za prebrojavanje crkava koje je sagradio svaki vladar.

Ocekivani rezultat: 5 vladara - Stefan Nemanja (2), ostali (1)

```
SELECT ?ime (COUNT(?c) AS ?brojCrkava) WHERE {
  ?v a nemanjicischema:Vladar ; nemanjicischema:Osoba.ime
    ?ime .
  ?c a nemanjicischema:Crkva ;
    nemanjicischema:Crkva.ktitor ?v .
} GROUP BY ?ime ORDER BY DESC(?brojCrkava)
```

ime	brojCrkava
"Stefan Nemanja"	"2"
"Stefan Decanski"	"1"
"Stefan Milutin"	"1"
"Stefan Prvovencani"	"1"
"Stefan Uros I"	"1"

Upit 05.rq: Lanac nasljedjivanja

Opis: Prikazuje parove vladar-nasljednik. Demonstrira rekurzivnu self-referencu klase Vladar.

Ocekivani rezultat: 9 parova nasljedjivanja

```
SELECT ?vladarIme ?nasljednikIme WHERE {
  ?v a nemanjicischema:Vladar ;
    nemanjicischema:Osoba.ime ?vladarIme ;
    nemanjicischema:Vladar.nasljednik ?n .
  ?n nemanjicischema:Osoba.ime ?nasljednikIme .
} ORDER BY ?vladarIme
```

vladarIme	nasljednikIme
"Stefan Decanski"	"Stefan Dusan"
"Stefan Dragutin"	"Stefan Milutin"
"Stefan Dusan"	"Stefan Uros V"
"Stefan Milutin"	"Stefan Decanski"
"Stefan Nemanja"	"Stefan Prvovencani"
"Stefan Prvovencani"	"Stefan Radoslav"
"Stefan Radoslav"	"Stefan Vladislav"
"Stefan Uros I"	"Stefan Dragutin"
"Stefan Vladislav"	"Stefan Uros I"

Upit 06.rq: Bitke sa pobjednicima (OPTIONAL)

Opis: Koristi OPTIONAL klauzulu jer neke bitke nemaju definisanog pobjednika (Bitka na Marici).

Ocekivani rezultat: 3 bitke, od kojih 1 nema pobjednika

```
SELECT ?naziv ?godina ?lokacija ?pobjednikIme WHERE {
  ?b a nemanjicischema:Bitka ;
  nemanjicischema:Bitka.naziv ?naziv ;
  nemanjicischema:Bitka.godina ?godina .
  OPTIONAL {
    ?b nemanjicischema:Bitka.pobjednik ?p .
    ?p nemanjicischema:Osoba.ime ?pobjednikIme .
  }
} ORDER BY ?godina
```

naziv	godina	lokacija	pobjednikIme
"Osvajanje Skoplja"	"1282"	"Skoplje"	"Stefan Milutin"
"Bitka na Velbudu"	"1330"	"Velbud (Kjustendil)"	"Stefan Decanski"
"Bitka na Marici"	"1371"	"rijeka Marica"	-

Upit 07.rq: Pravni dokumenti sa donosiocima

Opis: Spaja PravniDokument sa Vladar klasom. Prikazuje naziv dokumenta, godinu i ko ga je donio.

Ocekivani rezultat: 2 dokumenta (Povelja Nemanje 1196, Dusanov Zakonik 1349)

```
SELECT ?naziv ?godina ?opis ?donosilacIme WHERE {
  ?d a nemanjicischema:PravniDokument ;
  nemanjicischema:PravniDokument.naziv ?naziv ;
  nemanjicischema:PravniDokument.donosilac ?v .
  ?v nemanjicischema:Osoba.ime ?donosilacIme .
} ORDER BY ?godina
```

naziv	godina	opis	donosilacIme
"Povelja Stefana Nemanje"	"1196"	"Povelja kojom Stefan Nemanja daruje manastir Hilandar"	"Stefan Nemanja"
"Dusanov Zakonik"	"1349"	"Zakonik cara Stefana Dusana, prvi srpski zakonik"	"Stefan Dusan"

Upit 08.rq: M:N relacija - vladari sa vise od 2 teritorije

Opis: Demonstrira M:N relaciju Vladar-Teritorija. Koristi GROUP BY i HAVING za filtriranje po broju.

Ocekivani rezultat: Stefan Dusan (5 teritorija), Stefan Milutin (3), Stefan Nemanja (3)


```
SELECT ?ime (COUNT(?t) AS ?brojTerit) WHERE {
  ?v a nemanjicischema:Vladar ;
  nemanjicischema:Osoba.ime ?ime ;
  nemanjicischema:Vladar.vladaoTeritorijom ?t .
} GROUP BY ?ime HAVING (COUNT(?t) > 2)
ORDER BY DESC(?brojTerit)
```

ime	brojTerit
"Stefan Dusan"	"5"
"Stefan Milutin"	"3"
"Stefan Nemanja"	"3"

Upit 09.rq: Crkve na Kosovu

Opis: Filtrira crkve po specifičnoj lokaciji (ter_kosovo). Spaja Crkva, Vladar i Teritorija.

Očekivani rezultat: 3 crkve na Kosovu (Gracanica, Decani, Pecka Patrijarsija)

```
SELECT ?crkvaNaziv ?ktitorIme ?godina WHERE {
  ?c a nemanjicischema:Crkva ;
  nemanjicischema:Crkva.ktitor ?k ;
  nemanjicischema:Crkva.lokacija nemanjicischema:ter_kosovo .
  ?k nemanjicischema:Osoba.ime ?ktitorIme .
} ORDER BY ?godina
```

crkvaNaziv	ktitorIme	godina
"Pecka Patrijarsija"	"Stefan Uros I"	"1253"
"Manastir Gracanica"	"Stefan Milutin"	"1321"
"Manastir Decani"	"Stefan Decanski"	"1327"

Upit 10.rq: Svestenici i manastiri (OPTIONAL)

Opis: Prikazuje sveštenicke funkcije i manastire kojima upravljaju. OPTIONAL jer neki sveštenici mozda ne upravljaju manastirom.

Očekivani rezultat: 2 sveštenika (Sveti Sava -> Hilandar, Joanikije -> Pecka Patrijarsija)

```
SELECT ?svName ?funkcija ?manastir WHERE {
  ?s a nemanjicischema:Svestenik ;
  nemanjicischema:Osoba.ime ?svName ;
  nemanjicischema:Svestenik.crkvenaFunkcija ?funkcija .
  OPTIONAL {
    ?s nemanjicischema:Svestenik.upravljaManastirom ?c .
    ?c nemanjicischema:Crkva.naziv ?manastir .
  }
}
```

svName	funkcija	manastir
"Sveti Sava"	"Arhiepiskop"	"Manastir Hilandar"
"Joanikije II"	"Patrijarh"	"Pecka Patrijarsija"

5. Vanjski SPARQL upiti (11.rq - 20.rq)

Vanjskih 10 upita koristi Wikidata kao izvor podataka. Svi upiti se izvrsavaju na <https://query.wikidata.org/> endpointu.

NAPOMENA: PODACI NA WIKIDATA NISU UVIJEK DOSTUPNI, POTPUNI, POUZDANI ILI NAJTACNIJI!!!

Wikidata - ključni identifikatori

Wikidata ID	Opis
wd:Q403	Srbija
wd:Q208188	Nemanjicka dinastija
wd:Q183180	Stefan Nemanja
wd:Q47624	Sveti Sava
wd:Q44613	Manastir (tip objekta)
wd:Q83413	Bitka na Kosovu (1389)
wdt:P53	Porodica/dinastija
wdt:P17	Država
wdt:P31	Instanca od (tip)
wdt:P569/P570	Datum rođenja/smrti

Upit 11.rq: Osnivaci srpskih manastira

Endpoint: <https://query.wikidata.org/>

Opis: Pretražuje Wikidata za sve manastire (Q44613) u Srbiji (Q403) i njihove osnivace (P112).

```
#title: Wikidata - Osnivaci srpskih manastira
#comment: Endpoint: https://query.wikidata.org/
PREFIX wd:  <http://www.wikidata.org/entity/>
PREFIX wdt:  <http://www.wikidata.org/prop/direct/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?manastirIme ?osnivacIme
WHERE {
    ?manastir wdt:P31 wd:Q44613 ;
              wdt:P17 wd:Q403 ;
              rdfs:label ?manastirIme .
    FILTER(lang(?manastirIme) = "en")
    OPTIONAL {
        ?manastir wdt:P112 ?osnivac .
        ?osnivac rdfs:label ?osnivacIme .
        FILTER(lang(?osnivacIme) = "en")
    }
}

ORDER BY ?manastirIme
LIMIT 20
```

Ajdanovac Monastery	
Babičko Monastery	
Banja Monastery	
Banjska Monastery	Stefan Milutin
Bavanište monastery	
Beočin monastery	
Bešenovo monastery	
Bjele Vode Monastery	
Blagoveštenje Monastery	
Blagoveštenje Rudničko Monastery	
Bošnjačko Monastery	

Upit 12.rq: Srpski vladari iz srednjeg vijeka

Endpoint: <https://query.wikidata.org/>

Opis: Pronalazi osobe (Q5) srpskog drzavljanstva koje su imale monarsku poziciju.

```
#title: Wikidata - Srpski vladari iz perioda Nemanjica (1166-1371)
#comment: Endpoint: https://query.wikidata.org/
PREFIX wd:    <http://www.wikidata.org/entity/>
PREFIX wdt:    <http://www.wikidata.org/prop/direct/>
PREFIX rdfs:   <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd:    <http://www.w3.org/2001/XMLSchema#>

SELECT DISTINCT ?ime ?rodjenje ?smrt
WHERE {
    ?vladar wdt:P31 wd:Q5 ;
            wdt:P27 wd:Q403 ;
            wdt:P39 ?pozicija ;
            rdfs:label ?ime .
    ?pozicija wdt:P31*/wdt:P27* wd:Q116 .
    FILTER(lang(?ime) = "en")
    OPTIONAL { ?vladar wdt:P569 ?rodjenje }
    OPTIONAL { ?vladar wdt:P570 ?smrt }
    FILTER(!BOUND(?rodjenje) || (?rodjenje >= "1100-01-01"^^xsd:dateTime && ?rodjenje <=
"1371-12-31"^^xsd:dateTime))
}
ORDER BY ?rodjenje
LIMIT 20
```

ime	rodjenje	smrt
Stefan Dragutin	1 January 1251	20 March 1316
Stefan Uroš III	1 January 1276	11 November 1331

Upit 13.rq: Bitke u kojima je ucestvovala Srbija

Endpoint: <https://query.wikidata.org/>

Opis: Trazi vojne sukobe (Q178561) vezane za Srbiju - ili kao zemlja domacin ili kroz ucesnike.

```
#title: Wikidata - Bitke u kojima je ucestvovala Srbija (1166-1371)
#comment: Endpoint: https://query.wikidata.org/
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

SELECT DISTINCT ?ime ?datum
WHERE {
    ?bitka wdt:P31 wd:Q178561 ;
        rdfs:label ?ime .
    FILTER(lang(?ime) = "en")
    { ?bitka wdt:P710 ?ucesnik . ?ucesnik wdt:P17 wd:Q403 . }
    UNION
    { ?bitka wdt:P17 wd:Q403 }
    OPTIONAL { ?bitka wdt:P585 ?datum }
    FILTER(!BOUND(?datum) || (?datum >= "1166-01-01"^^xsd:dateTime && ?datum <= "1371-12-31"^^xsd:dateTime))
}
ORDER BY ?datum
LIMIT 20
```

ime

datum

Battle of Čegar

Battle of Mišar

Battle of Koshare

Battle of Cer

Battle of Kolubara

Battles of Batočina and Jagodina

Battle of Šušaja

Battle of Varvarin

Upit 14.rq: Djeca Stefana Nemanje

Endpoint: <https://query.wikidata.org/>

Opis: Trazi sve osobe koje imaju Stefana Nemanju (Q183180) kao oca (P22). Vraca Svetog Savu, Stefana Prvoencani i Vukana.

```
#title: Wikidata - Djeca Stefana Nemanje
#comment: Endpoint: https://query.wikidata.org/
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>

SELECT DISTINCT ?ime ?rodjenje
WHERE {
    ?stefan rdfs:label "Stefan Nemanja"@en .
    ?dijete wdt:P22 ?stefan ;
        rdfs:label ?ime .
    FILTER(lang(?ime) = "en")
    OPTIONAL { ?dijete wdt:P569 ?rodjenje }
}
ORDER BY ?rodjenje
```

ime	rodjenje
Stefan the First-Crowned	1 January 1165
Vukan Nemanjić of Serbia	1 January 1165
Stefan the First-Crowned	1 January 1166
Saint Sava	1 January 1169

Upit 15.rq: Srpske srednjovekovne drzave

Endpoint: <https://query.wikidata.org/>

Opis: Trazi istorijske drzave (Q3024240) vezane za Srbiju. Prikazuje periode postojanja.

```
#title: Wikidata - Srpske drzave iz perioda Nemanjica
#comment: Endpoint: https://query.wikidata.org/
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd: <http://www.w3.org/2001/XMLSchema#>

SELECT DISTINCT ?ime ?pocetak ?kraj
WHERE {
    ?drzava wdt:P31 wd:Q3024240 ;
        wdt:P17 wd:Q403 ;
        rdfs:label ?ime .
    FILTER(lang(?ime) = "en")
    OPTIONAL { ?drzava wdt:P571 ?pocetak }
    OPTIONAL { ?drzava wdt:P576 ?kraj }
    FILTER(!BOUND(?pocetak) || (?pocetak >= "1100-01-01"^^xsd:dateTime && ?pocetak <= "1400-01-01"^^xsd:dateTime))
}
ORDER BY ?pocetak
LIMIT 20
```

ime	pocetak	kraj
Principality of Zeta	1 January 1360	1 January 1516

Upit 16.rq: Clanovi dinastije Nemanjica

Endpoint: <https://query.wikidata.org/>

Opis: Koristi P53 (porodica) da nadje sve clanove Nemanjicke dinastije (Q208188). Jedna od kljucnih Wikidata relacija za porodice.

```
#title: Wikidata - Clanovi dinastije Nemanjica
#comment: Endpoint: https://query.wikidata.org/
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>

SELECT DISTINCT ?ime ?rodjenGodina
WHERE {
    ?dinastija rdfs:label ?dynLabel .
    FILTER(CONTAINS(LCASE(STR(?dynLabel)),
        "nemanj") && lang(?dynLabel) = "en")
    ?osoba wdt:P53 ?dinastija ;
        rdfs:label ?ime .
    FILTER(lang(?ime) = "en")
    OPTIONAL {
        ?osoba wdt:P569 ?r .
        BIND(YEAR(?r) AS ?rodjenGodina)
    }
}
ORDER BY ?rodjenGodina
```

Stefan Vukanović Nemanjić	
Stefan Nemanja	1113
Stefan Nemanja	1114
Vukan Nemanjić of Serbia	1165
Stefan the First-Crowned	1165
Stefan the First-Crowned	1166
Stefan Radoslav of Serbia	1192
Stephen Vladislav I of Serbia	1198
Anastasia of Serbia	1200
Stephen Uroš I of Serbia	1220
Stephen Uroš I of Serbia	1223
Stefan Dragutin	1251
Stefan Milutin	1253
Elizabeth of Serbia	1270
Stefan Uroš III	1276
Stephen Vladislav II of Symria	1280
Stephen Constantine of Serbia	1282
Ana-Neda	1290
Theodora Smilets of Bulgaria	1290
Simonida	1294
Milica	1300

Upit 17.rq: Srpski vladari sa periodom vladavine

Endpoint: <https://query.wikidata.org/>

Opis: Koristi P39 (pozicija) za trazenje srpskih kraljeva. Prikazuje period od-do vladavine.

```
#title: Wikidata - Srpski vladari sa periodom vladavine
#comment: Endpoint: https://query.wikidata.org/
PREFIX wd: <http://www.wikidata.org/entity/>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>
PREFIX p: <http://www.wikidata.org/prop/>
PREFIX ps: <http://www.wikidata.org/prop/statement/>
PREFIX pq: <http://www.wikidata.org/prop/qualifier/>
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>

SELECT DISTINCT ?ime ?pocetak ?kraj
WHERE {
  VALUES ?vladar {
    wd:Q24445 wd:Q355911 wd:Q346960 wd:Q73810
    wd:Q290786 wd:Q319597 wd:Q314346 wd:Q152419
  }
  ?vladar rdfs:label ?ime .
  FILTER(lang(?ime) = "en")
  ?vladar p:P39 ?stmt .
  ?stmt ps:P39 ?pozicija .
  OPTIONAL { ?stmt pq:P580 ?pocetak }
  OPTIONAL { ?stmt pq:P582 ?kraj }
}
ORDER BY ?pocetak
```

ime	pocetak	kraj
Stefan Uroš III		
Stefan the First-Crowned	1 April 1196	1 January 1217
Stefan the First-Crowned	1 January 1217	1 January 1228
Stefan Dragutin	1 January 1276	1 January 1282
Stefan Dragutin	1 January 1282	1 January 1316

Upit 18.rq: Srpski manastiri na UNESCO listi

Endpoint: <https://query.wikidata.org/>

Opis: Traži manastire (Q44613) u Srbiji (Q403) koji imaju UNESCO ID (P757). Vraca 186 rezultata.

```
#title: Wikidata - Srpski manastiri na UNESCO listi iz perioda Nemanjica
#comment: Endpoint: https://query.wikidata.org/
PREFIX wd:    <http://www.wikidata.org/entity/>
PREFIX wdt:    <http://www.wikidata.org/prop/direct/>
PREFIX rdfs:   <http://www.w3.org/2000/01/rdf-schema#>
PREFIX xsd:    <http://www.w3.org/2001/XMLSchema#>

SELECT DISTINCT ?ime ?unescoId ?godinaGradnje
WHERE {
    ?manastir wdt:P31 wd:Q44613 ;
              wdt:P17 wd:Q403 ;
              rdfs:label ?ime .
    FILTER(lang(?ime) = "en")
    OPTIONAL { ?manastir wdt:P757 ?unescoId }
    OPTIONAL {
        ?manastir wdt:P571 ?datum .
        BIND(YEAR(?datum) AS ?godinaGradnje)
    }
    FILTER(!BOUND(?datum) || (?datum >= "1166-01-01"^^xsd:dateTime && ?datum <= "1371-12-31"^^xsd:dateTime))
}
ORDER BY ?godinaGradnje
```

ime	unescoId	godinaGradnje
Radošin Monastery		
Vranjaš monastery		
Plužac Monastery		
St. Paraskevi Monastery, Berkasovo		
Grište Monastery		
Babičko Monastery		
Voljavča Monastery (Bresnica)		

Upit 19.rq: Detalji o Svetom Savi

Endpoint: <https://query.wikidata.org/>

Opis: Direktan upit za entitet Sveti Sava (Q47624). Traži sve instance od (P31) vrijednosti.

```
#title: Wikidata - Detalji o Svetom Savi
#comment: Endpoint: https://query.wikidata.org/
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>

SELECT ?svojstvo ?vrijednost
WHERE {
    ?sveti rdfs:label "Saint Sava"@en .
    {
        BIND("Mjesto rodjenja" AS ?svojstvo)
        ?sveti wdt:P19 ?v .
        ?v rdfs:label ?vrijednost .
        FILTER(lang(?vrijednost) = "en")
    } UNION {
        BIND("Otac" AS ?svojstvo)
        ?sveti wdt:P22 ?v .
        ?v rdfs:label ?vrijednost .
        FILTER(lang(?vrijednost) = "en")
    } UNION {
        BIND("Drzavljanstvo" AS ?svojstvo)
        ?sveti wdt:P27 ?v .
        ?v rdfs:label ?vrijednost .
        FILTER(lang(?vrijednost) = "en")
    } UNION {
        BIND("Zanimanje" AS ?svojstvo)
        ?sveti wdt:P106 ?v .
        ?v rdfs:label ?vrijednost .
        FILTER(lang(?vrijednost) = "en")
    } UNION {
        BIND("Datum rodjenja" AS ?svojstvo)
        ?sveti wdt:P569 ?vrijednost .
    } UNION {
        BIND("Datum smrti" AS ?svojstvo)
        ?sveti wdt:P570 ?vrijednost .
    }
}
```

svojstvo	vrijednost
Datum smrti	21 January 1235
Datum smrti	21 January 1236
Datum rodjenja	1 January 1169
Datum rodjenja	1 January 1174
Datum rodjenja	1 January 1175
Datum rodjenja	1 January 1176
Otac	Stefan Nemanja
Mjesto rodjenja	Stari Ras
Drzavljanstvo	Grand Principality of Serbia
Drzavljanstvo	Kingdom of Serbia
Zanimanje	writer
Zanimanje	Eastern Orthodox priest
Zanimanje	diplomat
Zanimanje	hagiographer

Upit 20.rq: Ucesnici Bitke na Kosovu 1389

Endpoint: <https://query.wikidata.org/>

Opis: Trazi osobe koje su ucestvovalе (P607) u Bici na Kosovu (Q83413). Vraca Lazara, Milosa Obilica i druge.

```
#title: Wikidata - Srpski ucesnici Bitke na Kosovu 1389
#comment: Endpoint: https://query.wikidata.org/
PREFIX rdfs: <http://www.w3.org/2000/01/rdf-schema#>
PREFIX wdt: <http://www.wikidata.org/prop/direct/>

SELECT DISTINCT ?ime
WHERE {
    ?bitka rdfs:label "Battle of Kosovo"@en .
    ?osoba wdt:P607 ?bitka ;
        wdt:P27 ?drzava ;
        rdfs:label ?ime .
    FILTER(lang(?ime) = "en")
    ?drzava rdfs:label ?drzavaNaziv .
    FILTER(CONTAINS(LCASE(STR(?drzavaNaziv)), "serbia") && lang(?drzavaNaziv) = "en")
}
ORDER BY ?ime
```

ime

Lazar Hrebeljanović

6. Alati i pokretanje

6.1 Enterprise Architect

Enterprise Architect (EA) je koriscen za kreiranje UML class dijagrama. Dijagram sadrzi 12 klasa, 3 enumeracije, relacije nasljedjivanja i asocijacione veze. EA projekat je sacuvan kao .gea fajl.

6.2 SPARQL Playground (lokalni upiti)

Za pokretanje lokalnih upita (01-10) koristi se SPARQL Playground:

1. Skinuti sparql-playground-master sa: <https://github.com/calipho-sib/sparql-playground>
2. Instalirati JDK 21 i podesiti PATH (jdk-21.0.1/bin)
3. Kopirati ontology.ttl, resource.ttl i prefixes.ttl u default/ttl-data/
4. Kopirati 01.rq - 10.rq u default/queries/
5. Pokrenuti start.bat -> otici na <http://localhost:8888>
6. Odabrati upit iz liste i kliknuti Run

Vazno: Koristiti sparql-playground-master verziju (Spring Boot 2.7.9) koja je kompatibilna sa Java 21. Starija verzija 1.5.1 nije kompatibilna.

6.3 Wikidata SPARQL Editor (vanjski upiti)

Za pokretanje vanjskih upita (11-20):

1. Otici na <https://query.wikidata.org/>
2. Zalijepiti sadrzaj jednog od upita (11.rq - 20.rq)
3. Kliknuti plavo dugme Run (Ctrl+Enter)
4. Rezultati se prikazuju kao tabela ispod editora

6.4 Online provjera TTL fajlova

Za validaciju TTL fajlova bez lokalnog alata, koristiti Python rdflib biblioteku ili online alate. Napomena: easyrdf.org ne podrzava tacke u nazivima resursa (npr. `nemanjicischema:TipVladara.Car`), ali to je validna Turtle sintaksa koja radi u svim standardnim SPARQL engineima.

7. RDF/XML schema fajl

NemanjiciSchema-rdfs-augmented.xml je RDF/XML fajl koji opisuje ontologiju u formatu kompatibilnom sa EA i Garden projektom. Ovaj fajl sadrzi iste informacije kao ontology.ttl, samo u XML formatu umjesto Turtle formata.

Struktura XML fajla

Svaka klasa je opisana kao `rdf:Description` sa odgovarajucim tipom:

```
<?xml version="1.0"?>
<rdf:RDF xmlns:rdf="..."
  xmlns:rdfs="..."
  xmlns:cims="..."
  xml:base="http://iec.ch/TC57/NemanjiciSchema">

  <rdf:Description rdf:about="#Vladar">
    <rdf:type rdf:resource="rdfs:Class"/>
    <rdfs:label>#Vladar</rdfs:label>
    <rdfs:subClassOf rdf:resource="#Osoba"/>
  </rdf:Description>

  <rdf:Description rdf:about="#Vladar.titula">
    <rdf:type rdf:resource="rdf:Property"/>
    <rdfs:domain rdf:resource="#Vladar"/>
    <rdfs:range rdf:resource="#TipVladara"/>
    <cims:multiplicity rdf:resource="cims:M:1"/>
  </rdf:Description>
</rdf:RDF>
```

8. Napomene i zakljucak

Kompatibilnost fajlova

- ontology.ttl koristi ext: prefiks (ne cims:) za multiplicity i stereotype - uskladjeno sa Garden projektom
- Vrijednosti enumeracija u resource.ttl su stringovi ("Car", "Manastir") a ne URI reference
- xml: prefiks je uklonjen iz TTL fajlova jer je rezervisan u Turtle specifikaciji
- Tacke u nazivima resursa (TipVladara.Car) su validna Turtle sintaksa
- Lokalni SPARQL upiti koriste PREFIX nemanjicischema: koji odgovara prefiksu u TTL fajlovima

Sta je obuhvaceno projektom

- + 12 klasa u ontologiji (Osoba, Vladar, Svestenik, Plemic, Dinastija, Drzava, Teritorija, Bitka, PravniDokument, Ugovor, Crkva, Grad)
 - + 3 enumeracije (TipVladara sa 5 vrijednosti, TipTeritorije sa 5 vrijednosti, TipCrkve sa 2 vrijednosti)
 - + 1 M:N relacija (Vladar <-> Teritorija)
 - + 3 nasljedjivanja (Vladar, Svestenik, Plemic -> Osoba)
 - + 39 instanci u resource.ttl
 - + 10 lokalnih SPARQL upita (SELECT, FILTER, GROUP BY, HAVING, OPTIONAL, JOIN)
 - + 10 vanjskih Wikidata upita
 - + UML dijagram u Enterprise Architect
 - + RDF/XML schema fajl
-